[1] `eq1:=Dv=Vv*t-1/2*g*t^2 // Dv - vert position; Vv - initial vert velocity`

$$\mathrm{Dv} = (\mathrm{Vv} \cdot t - \frac{g \cdot t^2}{2}) \tag{1}$$

[2] `eq1a:=subst(eq1,Vv=Vb) // 45 deg launch angle`

$$\mathrm{Dv} = (\mathrm{Vb} \cdot t - \frac{g \cdot t^2}{2}) \tag{2}$$

[3] `eq2:=lhs(eq1a)=factor(rhs(eq1a))`

$$\mathrm{Dv} = (\frac{t(2 \cdot \mathrm{Vb} - t \cdot g)}{2}) \tag{3}$$

[4] `eq3:=t=subst(solve(eq1a,t),Dv=0)[0]`

$$t = (\frac{\mathrm{Vb} + \mathrm{Vb}}{g}) \tag{4}$$

[5] `assume(Vb>0)`

$$\mathrm{Vb} \tag{5}$$

[6] `eq4:=simplify(eq3) // time in the air`

$$t = (\frac{2 \cdot \mathrm{Vb}}{g}) \tag{6}$$

[7] `eq5:=Dh=Vh*t // Dh - horiz distance traveled`

$$\mathrm{Dh} = (\mathrm{Vh} \cdot t) \tag{7}$$

[8] `eq5a:=subst(eq5,Vh=Vb) // 45 deg launch angle`

$$\mathrm{Dh} = (\mathrm{Vb} \cdot t) \tag{8}$$

[9] `eq6:=subst(eq5a,eq4)`

$$\mathrm{Dh} = (\frac{\mathrm{Vb} \cdot 2 \cdot \mathrm{Vb}}{g}) \tag{9}$$

10 `eq7:=Vb=solve(eq6,Vb)[0]`

$$\mathrm{Vb} = (\frac{\sqrt{2}\sqrt{\mathrm{Dh} \cdot g}}{2}) \tag{10}$$

11 `vars01:=tran([Dh=25_ft,g=_g_]) // convert(_g__(ft*s^-2))`

$$\begin{pmatrix} \mathrm{Dh} = (25\_\mathrm{ft}) \\ g = (1\_\mathrm{g}\_) \end{pmatrix} \tag{11}$$

12 `eq8:=subst(eq7,vars01)`

$$\mathrm{Vb} = (\frac{\sqrt{2}\sqrt{(25\_\mathrm{ft} \cdot \_\mathrm{g}\_)}}{2}) \tag{12}$$

13 `eq8a:=lhs(eq8)=convert(usimplify(rhs(eq8))_(ft*s^-1))`

$$\mathrm{Vb} = \left(20.0543164171 \frac{\_\mathrm{ft}}{\_\mathrm{s}}\right) \tag{13}$$

14 `eq9:=Vl=Vb*sqrt(2)`

$$\mathrm{Vl} = (\mathrm{Vb}\sqrt{2}) \tag{14}$$

15 `eq10:=subst(eq9,eq8a)`

$$\mathrm{Vl} = \left(28.3610862611 \frac{\_\mathrm{ft}}{\_\mathrm{s}}\right) \tag{15}$$

16 `eq11:=Dl=1/2*Al*tl^2 // Dl - launcher travel; Al - launch acceleration; tl = time to launch`

$$\mathrm{Dl} = (\frac{\mathrm{Al} \cdot \mathrm{tl}^2}{2}) \tag{16}$$

17 `eq12:=Vl=Al*tl`

$$\mathrm{Vl} = (\mathrm{Al} \cdot \mathrm{tl}) \tag{17}$$

18 `eq13:=tl=solve(eq12,tl)[0]`

$$\mathrm{tl} = (\frac{\mathrm{Vl}}{\mathrm{Al}}) \tag{18}$$

19 `eq14:=simplify(subst(eq11,eq13))`

$$\mathrm{Dl} = (\frac{\mathrm{Vl}^2}{2 \cdot \mathrm{Al}}) \tag{19}$$

20 `eq15:=Al=solve(eq14,Al)[0]`

$$\mathrm{Al} = (\frac{\mathrm{Vl}^2}{2 \cdot \mathrm{Dl}}) \tag{20}$$

21 `eq16:=Dl=2*pi*Rl*angl/360 // Rl - radius launch arm; angl - launch arm sweep angle`

$$\mathrm{Dl} = (\frac{2 \cdot \pi \cdot \mathrm{Rl} \cdot \mathrm{angl}}{360}) \tag{21}$$

22 `vars02:=tran([ Rl=1_ft, angl=45])`

$$\begin{pmatrix} \mathrm{Rl} = (1\_\mathrm{ft}) \\ \mathrm{angl} = 45 \end{pmatrix} \tag{22}$$

23 `eq17:=subst(eq16,vars02)`

$$\mathrm{Dl} = (0.785398163397\_\mathrm{ft}) \tag{23}$$

24 `eq18:=eval(subst(eq15,[eq10,eq17]))`

$$\mathrm{Al} = \left(512.065886703 \frac{\_\mathrm{ft}^2 \cdot \_\mathrm{ft}^{-1.0}}{\_\mathrm{s}^2}\right) \tag{24}$$

25 `eq18a:=lhs(eq18)=convert(usimplify(rhs(eq18))_(ft*s^-2))`

$$\mathrm{Al} = \left(512.065886703 \frac{\_\mathrm{ft}}{\_\mathrm{s}^2}\right) \tag{25}$$

26 `eq19:=Fl=Mp*Al // Fl - launch average force; Mp - payload mass`

$$\mathrm{Fl} = (\mathrm{Mp} \cdot \mathrm{Al}) \tag{26}$$

27 `eq20:=subst(eq19,[ Mp=1_lb, eq18a ])`

$$\mathrm{Fl} = \left(512.065886703 \frac{\_\mathrm{ft} \cdot \_\mathrm{lb}}{\_\mathrm{s}^2}\right) \tag{27}$$

[28] `eq21:=lhs(eq20)=convert(rhs(eq20)_lbf)`

$$\mathrm{Fl} = (15.9119171679\_\mathrm{lbf}) \tag{28}$$

[29]